This Maple worksheet is denoted sirs3pr.
It is used to study an SIRS model with demography, for R0>1 and α1 large.
The deterministic version of the model shows damped oscillations toward an endemic infection level.
The particular model dealt here has a state space with 3 variables: S, I, and R, and the infection rate is "proper", with S+I+R-1 in the denominator.
Two things are done here.
First we derive an (approximate) expression for the angular frequency of the deterministic model oscillations, and after that we give a derivation of the moments of a diffusion approximation.
Ingemar Nåsell, KTH, Stockholm, 2012-10-03.

```
> restart;
  with(LinearAlgebra,Transpose,Eigenvalues,
  CharacteristicPolynomial);
  with(VectorCalculus,Jacobian);
  interface(imaginaryunit=II);
```

$$[Transpose, Eigenvalues, CharacteristicPolynomial]$$

$$[Jacobian]$$

$$I \tag{1}$$

The reason for changing the notation used for the imaginary unit is that "I" will be used below to denote the number of infected individuals.
The original transition rates are stored in the table transA:

```
> transA:=table([[1,0,0]=mu*N,[1,0,-1]=delta*R,[-1,1,0]=beta*S*I/
  (S+I+R-1),[-1,0,0]=mu*S,[0,-1,1]=gamma*I,[0,-1,0]=mu*I,[0,0,-1]
  =mu*R]);
```

$$transA := table\left(\left[[0,0,-1]=\mu R, [0,-1,0]=\mu I, [0,-1,1]=\gamma I, [-1,0,0]=\mu S, [-1,1,0]=\frac{\beta S I}{S+I+R-1}, [1,0,-1]=\delta R, [1,0,0]=\mu N\right]\right) \tag{2}$$

After scaling: x1=S/N, x2=I/N, x3=R/N, and reparametrization R0=β/(γ+μ), α1= (γ+μ)/μ, α2=(δ+μ)/μ, we get: S=x1*N, I=x2*N, R=x3*N, β=μ*α1*R0, γ=μ*(α1−1), δ=μ*(α2−1).
The Maple procedure "scale" is used to change the transition rates:

```
> scale:=proc(tab)
    local xA,n,xB,xC;
    xA:=op(2,eval(tab));
    n:=nops(xA);
    xB:=subs(S=x1*N,I=x2*N,R=x3*N,beta=mu*alpha1*R0,gamma=mu*
  (alpha1-1),delta=mu*(alpha2-1),xA);
    xC:=[seq(lhs(op(i,xB))=simplify(rhs(op(i,xB)/N)),i=1..n)];
    table(xC);
  end proc:
```

Apply the scaling and reparamerization described above to get the table of transition rates "trans":

```
> trans:=scale(transA);
```

$$trans := table\left(\left[[0,0,-1] = \mu\, x3, [0,-1,0] = \mu\, x2, [0,-1,1] = \mu\,(\alpha 1 - 1)\, x2, [-1,0,0] = \mu\, x1, [-1,1,0] = \frac{N\,\mu\,\alpha 1\, R0\, x1\, x2}{x1\, N + x2\, N + x3\, N - 1}, [1,0,-1] = \mu\,(\alpha 2 - 1)\, x3, [1,0,0] = \mu\right]\right) \quad (3)$$

Next is a procedure that determines the right-hand sides of the deterministic ODEs for the scaled variables x1, x2, x3 from the table of transition rates:

```
> equ:=proc(i,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The 3 right-hand sides are as follows:

```
> eq1:=equ(1,trans);
  eq2:=simplify(equ(2,trans));
  eq3:=equ(3,trans);
```

$$eq1 := -\mu\, x1 - \frac{N\,\mu\,\alpha 1\, R0\, x1\, x2}{x1\, N + x2\, N + x3\, N - 1} + \mu\,(\alpha 2 - 1)\, x3 + \mu$$

$$eq2 := \frac{\mu\, x2\, \alpha 1\,(-x1\, N - x2\, N - x3\, N + 1 + N\, R0\, x1)}{x1\, N + x2\, N + x3\, N - 1}$$

$$eq3 := -\mu\, x3 + \mu\,(\alpha 1 - 1)\, x2 - \mu\,(\alpha 2 - 1)\, x3 \quad (4)$$

Critical points:

```
> crit:=solve({eq1,eq2,eq3},{x1,x2,x3});
```

$$crit := \{x1 = 1, x2 = 0, x3 = 0\}, \left\{x1 = \frac{N-1}{N\, R0}, x2 = \frac{\alpha 2\,(-N + N\, R0 + 1)}{N\, R0\,(\alpha 2 + \alpha 1 - 1)}, x3 = \frac{(\alpha 1 - 1)\,(-N + N\, R0 + 1)}{N\, R0\,(\alpha 2 + \alpha 1 - 1)}\right\} \quad (5)$$

The point corresponding to an endemic infection level is termed (x10,x20,x30):

```
> x10:=rhs(crit[2][1]);
  x20:=rhs(crit[2][2]);
  x30:=map(factor,rhs(crit[2][3]));
```

$$x10 := \frac{N-1}{N\, R0}$$

$$x20 := \frac{\alpha 2\,(-N + N\, R0 + 1)}{N\, R0\,(\alpha 2 + \alpha 1 - 1)}$$

$$x30 := \frac{(\alpha 1 - 1)\,(-N + N\, R0 + 1)}{N\, R0\,(\alpha 2 + \alpha 1 - 1)} \quad (6)$$

The Jacobian of the system of ODEs is denoted Bx:

```
> Bx:=Jacobian([eq1,eq2,eq3],[x1,x2,x3]);
```

$$Bx := \left[\left[-\mu - \frac{N\mu\,\alpha 1\,R0\,x2}{x1\,N + x2\,N + x3\,N - 1} + \frac{N^2\,\mu\,\alpha 1\,R0\,x1\,x2}{(x1\,N + x2\,N + x3\,N - 1)^2},\right.\right.$$
$$-\frac{N\mu\,\alpha 1\,R0\,x1}{x1\,N + x2\,N + x3\,N - 1} + \frac{N^2\,\mu\,\alpha 1\,R0\,x1\,x2}{(x1\,N + x2\,N + x3\,N - 1)^2},$$
$$\left.\frac{N^2\,\mu\,\alpha 1\,R0\,x1\,x2}{(x1\,N + x2\,N + x3\,N - 1)^2} + \mu\,(\alpha 2 - 1)\right],$$
$$\left[\frac{\mu\,x2\,\alpha 1\,(-N + N\,R0)}{x1\,N + x2\,N + x3\,N - 1} - \frac{\mu\,x2\,\alpha 1\,(-x1\,N - x2\,N - x3\,N + 1 + N\,R0\,x1)\,N}{(x1\,N + x2\,N + x3\,N - 1)^2},\right.$$
$$\frac{\mu\,\alpha 1\,(-x1\,N - x2\,N - x3\,N + 1 + N\,R0\,x1)}{x1\,N + x2\,N + x3\,N - 1} - \frac{\mu\,x2\,\alpha 1\,N}{x1\,N + x2\,N + x3\,N - 1}$$
$$- \frac{\mu\,x2\,\alpha 1\,(-x1\,N - x2\,N - x3\,N + 1 + N\,R0\,x1)\,N}{(x1\,N + x2\,N + x3\,N - 1)^2}, -\frac{\mu\,x2\,\alpha 1\,N}{x1\,N + x2\,N + x3\,N - 1}$$
$$\left.- \frac{\mu\,x2\,\alpha 1\,(-x1\,N - x2\,N - x3\,N + 1 + N\,R0\,x1)\,N}{(x1\,N + x2\,N + x3\,N - 1)^2}\right],$$
$$\left.\left[0, \mu\,(\alpha 1 - 1), -\mu - \mu\,(\alpha 2 - 1)\right]\right] \quad \textbf{(7)}$$

Evaluate the Jacobian at the critical point:

```
> B:=simplify(subs(x1=x10,x2=x20,x3=x30,Bx));
```

$$B := \left[\left[-\frac{1}{R0\,(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)}\big(\mu\,(N\,\alpha 2\,\alpha 1 - 2\,N\,\alpha 1\,R0\,\alpha 2\right.\right.$$
$$+ \alpha 1\,N\,R0 + R0^2\,\alpha 2\,\alpha 1\,N - N\,R0 + N\,R0\,\alpha 2 - \alpha 2\,\alpha 1 - R0\,\alpha 1 + R0\,\alpha 2\,\alpha 1 + R0$$
$$- \alpha 2\,R0)\big), -\frac{\mu\,\alpha 1\,(\alpha 1\,N\,R0 - N\,R0 - \alpha 2\,R0 - R0\,\alpha 1 + R0 + \alpha 2\,N - \alpha 2)}{R0\,(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)},$$
$$\frac{1}{R0\,(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)}\big(\mu\,(2\,N\,\alpha 1\,R0\,\alpha 2 - \alpha 1\,N\,R0 - N\,\alpha 2\,\alpha 1$$
$$+ N\,R0 - 2\,N\,R0\,\alpha 2 + N\,R0\,\alpha 2^2 + \alpha 2\,\alpha 1 + R0\,\alpha 1 - R0\,\alpha 2\,\alpha 1 - R0 + 2\,\alpha 2\,R0$$
$$\left. - \alpha 2^2\,R0)\big)\right],$$
$$\left[\frac{\mu\,\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1\,(-1 + R0)}{R0\,(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)},\right.$$
$$-\frac{\mu\,\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1}{R0\,(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)}, -\frac{\mu\,\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1}{R0\,(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)}$$
$$\Big],\quad \textbf{(8)}$$

$$\left[0, \mu\,(\alpha 1-1), -\mu\,\alpha 2\right]\Big]$$

We proceed to determine the eigenvalues of the matrix B.
We use first the command "Eigenvalues" to show that one of the eigenvalues equals -μ.
After that, we use the command "CharacteristicPolynomial" and the knowledge that one eigenvalue equals −μ to derive a quadratic equation for the remaining two eigenvalues.

```
> eig:=Eigenvalues(B);
```

$$eig := \Big[[-\mu], \quad (9)$$

$$\Big[-\frac{1}{2}\,\frac{1}{\alpha 2\,N+\alpha 1\,N-N-\alpha 2-\alpha 1+1}\Big(\Big(N\,\alpha 2^2-\alpha 2^2+N\,\alpha 1\,R0\,\alpha 2-\alpha 2\,N$$
$$+\alpha 2$$
$$-\big(\alpha 2^2+16\,N\,\alpha 2^2\,\alpha 1+16\,N\,\alpha 2\,\alpha 1^2-8\,N\,\alpha 2\,\alpha 1-8\,\alpha 2\,\alpha 1^2\,N^2$$
$$-8\,\alpha 2^2\,\alpha 1\,N^2-2\,N\,\alpha 2^2-8\,\alpha 2^2\,\alpha 1-8\,\alpha 2\,\alpha 1^2+4\,\alpha 2\,\alpha 1+\alpha 2^2\,N^2$$
$$+4\,N\,\alpha 1\,R0\,\alpha 2-4\,N^2\,R0\,\alpha 2\,\alpha 1+6\,\alpha 2^2\,\alpha 1\,N^2\,R0+8\,\alpha 2\,\alpha 1^2\,N^2\,R0$$
$$-6\,\alpha 2^2\,\alpha 1\,N\,R0-8\,\alpha 2\,\alpha 1^2\,N\,R0+4\,\alpha 2\,N^2\,\alpha 1+4\,N\,\alpha 2^3-2\,\alpha 2^3\,N^2-2\,\alpha 2^3$$
$$-8\,N^2\,\alpha 2^2\,\alpha 1^2\,R0-2\,N^2\,\alpha 2^3\,\alpha 1\,R0+8\,N\,\alpha 2^2\,\alpha 1^2\,R0+2\,N\,\alpha 2^3\,\alpha 1\,R0$$
$$+N^2\,\alpha 1^2\,R0^2\,\alpha 2^2-4\,\alpha 2\,\alpha 1^3\,N^2\,R0+4\,\alpha 2\,\alpha 1^3\,N\,R0+4\,N^2\,\alpha 2^3\,\alpha 1-8\,N\,\alpha 2^3\,\alpha 1$$
$$+4\,\alpha 2\,\alpha 1^3\,N^2-8\,\alpha 2\,\alpha 1^3\,N+8\,\alpha 2^2\,\alpha 1^2+8\,N^2\,\alpha 2^2\,\alpha 1^2-16\,N\,\alpha 2^2\,\alpha 1^2$$
$$+4\,\alpha 2^3\,\alpha 1+4\,\alpha 2\,\alpha 1^3+\alpha 2^4+N^2\,\alpha 2^4-2\,N\,\alpha 2^4\big)^{1/2}\Big)\,\mu\Big)\Big],$$

$$\Big[-\frac{1}{2}\,\frac{1}{\alpha 2\,N+\alpha 1\,N-N-\alpha 2-\alpha 1+1}\Big(\Big(N\,\alpha 2^2-\alpha 2^2+N\,\alpha 1\,R0\,\alpha 2-\alpha 2\,N$$
$$+\alpha 2$$
$$+\big(\alpha 2^2+16\,N\,\alpha 2^2\,\alpha 1+16\,N\,\alpha 2\,\alpha 1^2-8\,N\,\alpha 2\,\alpha 1-8\,\alpha 2\,\alpha 1^2\,N^2$$
$$-8\,\alpha 2^2\,\alpha 1\,N^2-2\,N\,\alpha 2^2-8\,\alpha 2^2\,\alpha 1-8\,\alpha 2\,\alpha 1^2+4\,\alpha 2\,\alpha 1+\alpha 2^2\,N^2$$
$$+4\,N\,\alpha 1\,R0\,\alpha 2-4\,N^2\,R0\,\alpha 2\,\alpha 1+6\,\alpha 2^2\,\alpha 1\,N^2\,R0+8\,\alpha 2\,\alpha 1^2\,N^2\,R0$$
$$-6\,\alpha 2^2\,\alpha 1\,N\,R0-8\,\alpha 2\,\alpha 1^2\,N\,R0+4\,\alpha 2\,N^2\,\alpha 1+4\,N\,\alpha 2^3-2\,\alpha 2^3\,N^2-2\,\alpha 2^3$$
$$-8\,N^2\,\alpha 2^2\,\alpha 1^2\,R0-2\,N^2\,\alpha 2^3\,\alpha 1\,R0+8\,N\,\alpha 2^2\,\alpha 1^2\,R0+2\,N\,\alpha 2^3\,\alpha 1\,R0$$

$$+ N^2\,\alpha 1^2\,R0^2\,\alpha 2^2 - 4\,\alpha 2\,\alpha 1^3\,N^2\,R0 + 4\,\alpha 2\,\alpha 1^3\,N\,R0 + 4\,N^2\,\alpha 2^3\,\alpha 1 - 8\,N\,\alpha 2^3\,\alpha 1$$

$$+ 4\,\alpha 2\,\alpha 1^3\,N^2 - 8\,\alpha 2\,\alpha 1^3\,N + 8\,\alpha 2^2\,\alpha 1^2 + 8\,N^2\,\alpha 2^2\,\alpha 1^2 - 16\,N\,\alpha 2^2\,\alpha 1^2$$

$$+ 4\,\alpha 2^3\,\alpha 1 + 4\,\alpha 2\,\alpha 1^3 + \alpha 2^4 + N^2\,\alpha 2^4 - 2\,N\,\alpha 2^4)^{1/2}\Big)\,\mu\Big)\Big]\Big]$$

One of the eigenvalues is thus seen to be equal to $-\mu$.
Next we determine the characteristic polynomial:

```
> p:=CharacteristicPolynomial(B,lambda);
```

$$p := \lambda^3 + \frac{\mu\,(N\,\alpha 2^2 - \alpha 2^2 + \alpha 1\,N + N\,\alpha 1\,R0\,\alpha 2 - N - \alpha 1 + 1)\,\lambda^2}{\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1} + \frac{1}{\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1}\,\big(\alpha 2\,\mu^2\,(-N\,\alpha 1^2 + \alpha 1\,N - N\,\alpha 2\,\alpha 1 + N\,\alpha 1\,R0\,\alpha 2 - N + \alpha 2\,N - \alpha 1 + \alpha 2\,\alpha 1 + 1 - \alpha 2 + R0\,N\,\alpha 1^2 + \alpha 1^2)\,\lambda\big) + \frac{\mu^3\,\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1}{N - 1} \tag{10}$$

To proceed, we derive a quadratic equation for the remaining two eigenvalues by dividing p by $\lambda+\mu$, and simplifying:

```
> p1:=map(simplify,collect(simplify(p/(lambda+mu)),lambda));
```

$$p1 := \lambda^2 + \frac{\mu\,\alpha 2\,(\alpha 2\,N - \alpha 2 + \alpha 1\,N\,R0 - N + 1)\,\lambda}{\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1} + \frac{\mu^2\,\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1}{N - 1} \tag{11}$$

The two roots of p1 = 0, where p1 = $\lambda^2 + a\cdot\lambda$ + b, can be written $-\frac{a}{2} \pm i\cdot\Omega$, where $\Omega = \sqrt{b - \left(\frac{a}{2}\right)^2}$.

We note that μ is a factor in Ω, so we can write $\Omega = \mu\cdot\sqrt{b1 - \left(\frac{a1}{2}\right)^2}$, where $b1 = \frac{b}{\mu^2}$, and $a1 = \frac{a}{\mu}$.

.Thus,

```
> a1:=op(2,p1)/mu/lambda;
  b1:=op(3,p1)/mu^2;
```

$$a1 := \frac{\alpha 2\,(\alpha 2\,N - \alpha 2 + \alpha 1\,N\,R0 - N + 1)}{\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1}$$

$$b1 := \frac{\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1}{N - 1} \tag{12}$$

Consider the quantity C = b1 - $\left(\frac{a1}{2}\right)^2$

```
> C:=b1-(a1/2)^2;
```

(13)

$$C := \frac{\alpha 2\,(-N + N\,R0 + 1)\,\alpha 1}{N-1} - \frac{1}{4}\,\frac{\alpha 2^2\,(\alpha 2\,N - \alpha 2 + \alpha 1\,N\,R0 - N + 1)^2}{(\alpha 2\,N + \alpha 1\,N - N - \alpha 2 - \alpha 1 + 1)^2} \quad \textbf{(13)}$$

C is seen to depend on N.
We determine the asymptotic approximation of C as N becomes large.
The one-term asymptotic approximation of C, denoted C1, equals the limit of C as N approaches infinity.

```
> C1:=limit(b1,N=infinity)-limit((a1/2)^2,N=infinity);
```

$$C1 := \alpha 2\,(-1 + R0)\,\alpha 1 - \frac{1}{4}\,\frac{\alpha 2^2\,(\alpha 2 + R0\,\alpha 1 - 1)^2}{(\alpha 2 + \alpha 1 - 1)^2} \quad \textbf{(14)}$$

This can be written

```
> C2:=op(1,C1)-(alpha2*R1/2)^2;
```

$$C2 := \alpha 2\,(-1 + R0)\,\alpha 1 - \frac{1}{4}\,\alpha 2^2\,R1^2 \quad \textbf{(15)}$$

where

```
> R1=(alpha1*R0+alpha2-1)/(alpha1+alpha2-1);
```

$$R1 = \frac{\alpha 2 + R0\,\alpha 1 - 1}{\alpha 2 + \alpha 1 - 1} \quad \textbf{(16)}$$

We summarize: The limit of Ω as N approaches infinity can be written $\Omega 1 = \mu^* \sqrt{\alpha 1 \cdot \alpha 2 \cdot (R0 - 1) - \left(\alpha 2 \cdot \frac{R1}{2}\right)^2}$

This finishes the study of the eigenvalues.
The rest of the worksheet is used to derive approximations of the matrix of covariances for the diffusion approximation.
Covariances of x[i]x[j] are determined by cov1:

```
> cov1:=proc(i,j,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*lhs(x[n])[j]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The local covariance matrix S is determined by the procedure cov:

```
> cov:=proc(tab)
    local i,j,d,S;
    d:=nops(lhs(op(2,eval(tab))[1]));
    for i from 1 to d do
      for j from 1 to d do
        S[i,j]:=cov1(i,j,tab);
      od;
    od;
    S:=Matrix(d,S);
  end proc:
```

By using the table of transition rates in "trans", we get

```
> Sx:=simplify(cov(trans));
```

(17)

$$Sx := \left[\left[\frac{1}{x1\,N + x2\,N + x3\,N - 1}\left(\mu\left(x1^2\,N + x1\,x2\,N - x1 + N\,\alpha 1\,R0\,x1\,x2 + x3\,\alpha 2\,x1\,N + x3\,\alpha 2\,x2\,N + \alpha 2\,x3^2\,N - x3\,\alpha 2 - x3\,x2\,N - x3^2\,N + x3 + x1\,N + x2\,N + x3\,N - 1\right)\right), -\frac{N\,\mu\,\alpha 1\,R0\,x1\,x2}{x1\,N + x2\,N + x3\,N - 1}, -\mu\left(\alpha 2 - 1\right)x3\right], \left[-\frac{N\,\mu\,\alpha 1\,R0\,x1\,x2}{x1\,N + x2\,N + x3\,N - 1}, \frac{\mu\,x2\,\alpha 1\left(x1\,N + x2\,N + x3\,N - 1 + N\,R0\,x1\right)}{x1\,N + x2\,N + x3\,N - 1}, -\mu\left(\alpha 1 - 1\right)x2\right], \left[-\mu\left(\alpha 2 - 1\right)x3, -\mu\left(\alpha 1 - 1\right)x2, \alpha 1\,\mu\,x2 - \mu\,x2 + \mu\,x3\,\alpha 2\right]\right] \quad \textbf{(17)}$$

Evaluate the local covariance matrix at the critical point:

```
> S:=simplify(subs(x1=x10,x2=x20,x3=x30,Sx));
```

$$S := \left[\left[\frac{2\,\mu\left(-N\,\alpha 2\,\alpha 1 + \alpha 1\,N + N\,\alpha 1\,R0\,\alpha 2 - N + \alpha 2\,N - \alpha 1 + \alpha 2\,\alpha 1 + 1 - \alpha 2\right)}{R0\left(\alpha 2 + \alpha 1 - 1\right)N}, -\frac{\alpha 1\,\mu\,\alpha 2\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}, -\frac{\mu\left(\alpha 2 - 1\right)\left(\alpha 1 - 1\right)\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}\right], \left[-\frac{\alpha 1\,\mu\,\alpha 2\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}, \frac{2\,\alpha 1\,\mu\,\alpha 2\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}, -\frac{\mu\left(\alpha 1 - 1\right)\alpha 2\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}\right], \left[-\frac{\mu\left(\alpha 2 - 1\right)\left(\alpha 1 - 1\right)\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}, -\frac{\mu\left(\alpha 1 - 1\right)\alpha 2\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}, \frac{2\,\mu\left(\alpha 1 - 1\right)\alpha 2\left(-N + N\,R0 + 1\right)}{N\,R0\left(\alpha 2 + \alpha 1 - 1\right)}\right]\right] \quad \textbf{(18)}$$

Now proceed to solve A=-S, where A=B*SIG+SIG*BT, and where BT=Transpose(B).
First introduce notation for the elements of the matrix SIG:

```
> SIG:=Matrix(3,[s11,s12,s13,s21,s22,s23,s31,s32,s33]);
```

$$SIG := \begin{bmatrix} s11 & s12 & s13 \\ s21 & s22 & s23 \\ s31 & s32 & s33 \end{bmatrix} \quad \textbf{(19)}$$

Next, evaluate the matrix A:

```
> A:=Matrix(evalm(B&*SIG+SIG&*Transpose(B))):
```

Solve the 9 scalar equations that result from the matrix equation A+S=0 for the 9 unknowns in SIG:

```
> solve(convert(A+S,set),convert(SIG,set)):
```

```
> assign(%);
```

All the elements of the matrix SIG depend on both N and on α1.
We determine the one-term asymptotic approximations of each of them as both N and α1 become large:

```
> s11a:=simplify(op(1,asympt(op(1,asympt(s11,alpha1)),N)));
  s12a:=simplify(op(1,asympt(op(1,asympt(s12,alpha1)),N)));
  s13a:=simplify(op(1,asympt(op(1,asympt(s13,alpha1)),N)));
  s21a:=simplify(op(1,asympt(op(1,asympt(s21,alpha1)),N)));
  s22a:=simplify(op(1,asympt(op(1,asympt(s22,alpha1)),N)));
  s23a:=simplify(op(1,asympt(op(1,asympt(s23,alpha1)),N)));
  s31a:=simplify(op(1,asympt(op(1,asympt(s31,alpha1)),N)));
  s32a:=simplify(op(1,asympt(op(1,asympt(s32,alpha1)),N)));
  s33a:=simplify(op(1,asympt(op(1,asympt(s33,alpha1)),N)));
```

$$s11a := \frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s12a := -\frac{1}{R0}$$

$$s13a := -\frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s21a := -\frac{1}{R0}$$

$$s22a := \frac{-1 + R0}{R0^2}$$

$$s23a := \frac{1}{R0^2}$$

$$s31a := -\frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s32a := \frac{1}{R0^2}$$

$$s33a := \frac{\alpha 1}{\alpha 2\, R0^2} \qquad \textbf{(20)}$$

The Expectation of S+I+R, divided by N, is denoted Expsum:

```
> Expsum:=simplify(x10+x20+x30);
```

$$Expsum := 1 \qquad \textbf{(21)}$$

The Variance of S+I+R, divided by N, is denotede Varsum:

```
> Varsum:=simplify(s11+s22+s33+2*s12+2*s13+2*s23);
```

$$Varsum := 1 \qquad \textbf{(22)}$$

I define ρI as the first term in the asymptotic approximation of the ratio x20*N/ $\sqrt{N \cdot s22}$ for large α1 and N.

```
> x20;
  x20a:=op(1,asympt(op(1,asympt(x20,alpha1)),N));
```

```
rhoI:=simplify(x20a*N/sqrt(s22a*N)) assuming R0>1;
```

$$\frac{\alpha 2\,(-N+N\,R0+1)}{N\,R0\,(\alpha 2+\alpha 1-1)}$$

$$x20a := \frac{\alpha 2\,(-1+R0)}{R0\,\alpha 1}$$

$$rhoI := \frac{\alpha 2\sqrt{-1+R0}\,\sqrt{N}}{\alpha 1} \quad (23)$$

This expression for rhoI is the same as for sirs2c and sirs3c.